

University of Lleida

Master's Degree in Informatics Engineering

Higher Polythecnic School

Data Producer 1

Ubiquitous Computing and Embedded Systems

Francesc Contreras

Albert Pérez

Marc Visa

November 10, 2021

Table of contents

List of Figures

1 Introduction

The purpose of this document is to explain the Data Producer 1 development for the Sprint 1 of the project.

We shall introduce all the steps and software we have used when developing the it and showing images as a real example for detailed explanation.

2 Environment

At first, these are the steps for installing the libraries and setting up the Arduino IDE to get started for the Data Producer 1 development:

1. Install Arduino IDE v.1.8.16.
2. Install driver for USB adapter to ESP-01 and switch the interrupter to PROG mode.
3. Install ESP8266 Board from Boards Manager from Arduino IDE.
4. Install the following libraries from the Manage Libraries of Arduino IDE:
   (a) DHT [1] library by Adafruit.
   (b) Adafruit Unified Sensor by Adafruit.
5. Install the following libraries from GitHub: (These libraries should be added to the Arduino/libraries folder):
   (a) PubSubClient[2]
6. Connect the ESP-01[3] along with the USB adapter to your machine.
7. Code the program for the ESP-01 and DHT11.

[1] Temperature and humidity sensor
[2] MQTT Library
[3] Low-cost WiFi microchip with built-in TCP/IP networking software, and microcontroller capability



Figure 1: Installing the DHT sensor library for ESPx from the Arduino Manage Libraries.



Figure 2: Installing the Adafruit Unified Sensor from the Arduino Manage Libraries.



Figure 3: Installing the ESP8266 Board from the Arduino Boards Manager.

2.1 Material required

Subsequently, you will find all the required components for the Data Producer 1, and a short explanatory description for those not so common.

- **ESP-01 (3.3 V)**: is a Wi-Fi module that allows microcontrollers access to a Wi-Fi network.
- **Breadboard**: A board, having a matrix of small holes to which components may be attached without solder.
- **LCD Screen (5V)**: is a flat-panel display or other electronically modulated optical device that uses the light-modulating properties of liquid crystals combined with polarizers.
- **DHT11 (3.3V)**: is a basic, ultra low-cost digital temperature and humidity sensor.
- **VCC Protoboard adapter (3.3V, 5V)**
- **ESP Programmer module**
- **Wires**

3 Development

Regards the development, this is based on two parts, the first one realated to work directly with the electronic components, and the second by programming the arduino sketch.

The following sections shows you how the electronic schema is in real, as well as how finally we got the visualization of the temperature and the humidity by two different ways.

Moreover, there is the code shown below, but also you may consult the github repository of the project, the one is specified subsequently.

3.1 Hardware

3.1.1 Diagram



Figure 4: Data Producer 1 Diagram

Pay Attention!, in our LCD Screen we don't have SDA and SCL connections, we have RX and TX, then our RX is SDA and TX is SCL in the Figure 4.

3.1.2 Assembly



Figure 5: Data Producer 1 Assembly

3.2 Software

```
// Import required libraries
#include <Arduino.h>
#include <ESP8266WiFi.h>

#include <Adafruit_Sensor.h>
#include <DHT.h>

//MQTT Pub
#include <PubSubClient.h>

// Replace with your network credentials
const char* ssid = "-";
const char* password = "-";

// DHT11
#define DHTPIN 2   // Digital pin connected to the DHT sensor
#define DHTTYPE  DHT11    // DHT 11

DHT dht(DHTPIN, DHTTYPE);

float t = 0.0;
float h = 0.0;

// MQTT

const char* mqtt_server = "192.168.1.163"; // MQTT BROKER IP
WiFiClient espClient;
```

```
PubSubClient client(espClient);

#define MSG_BUFFER_SIZE (50)
char msg[MSG_BUFFER_SIZE];

void callback(char* topic, byte* payload, unsigned int length) {
  /*Serial.print("Message arrived [");
  Serial.print(topic);
  Serial.print("] ");
  for (int i=0;i<length;i++) {
    Serial.print((char)payload[i]);
  }
  Serial.println();*/
}

void reconnect() {
  // Loop until we're reconnected
  while (!client.connected()) {
    Serial.print("Attempting MQTT connection...");
    // Create a random client ID
    String clientId = "DataProducer1";
    clientId += String(random(0xffff), HEX);
    // Attempt to connect
    if (client.connect(clientId.c_str())) {
      Serial.println("connected");
      // Once connected, publish an announcement...
      client.publish("outTopic", "connected");
      // ... and resubscribe
      client.subscribe("#");
    } else {
      Serial.print("failed, rc=");
      Serial.print(client.state());
      Serial.println(" try again in 5 seconds");
      // Wait 5 seconds before retrying
      delay(5000);
    }
  }
}

void setup(){

  Serial.begin(115200);

  // Connect to Wi-Fi
  WiFi.begin(ssid, password);
  Serial.println("Connecting to WiFi");

  while (WiFi.status() != WL_CONNECTED) {
    delay(1000);
    Serial.println(".");
  }

  // Print ESP8266 Local IP Address
```

```
  Serial.println(WiFi.localIP());

  dht.begin();

  client.setServer(mqtt_server, 1883);
  client.setCallback(callback);
}

void clean_screen()
{
  Serial.write(0x7C);
  Serial.write(0x00);
}

void loop(){
    clean_screen();

    // MQTT
      if (!client.connected()) {
        reconnect();
      }
      client.loop();

    // Temperature
    float newT = dht.readTemperature();
    if (isnan(newT)) {
      Serial.println("Failed to read from DHT sensor!");
    }
    else {
      t = newT;
      String outStringT = (String) t+"C ";
      Serial.print(outStringT);
      snprintf (msg, MSG_BUFFER_SIZE, "%f C", t);
      client.publish("/temperature", msg);
    }

    // Humidity
    float newH = dht.readHumidity();
    // if humidity read failed, don't change h value
    if (isnan(newH)) {
      Serial.println("Failed to read from DHT sensor!");
    }
    else {
      h = newH;
      String outStringH = (String) h+"%";
      Serial.print(outStringH);
      snprintf (msg, MSG_BUFFER_SIZE, "%f ", h);
      client.publish("/humidity", msg);
    }

    delay(5000);
}
```

4 Code repository

Wind Turbine Generator Github repository

References

[1] ESP8266 DHT11/DHT22 Temperature and Humidity Web Server with Arduino IDE